
***(

Recall that in Lectures 5-6 we saw that lambda expressions
could be compiled into simpler-to-implement CL terms,
and that this was a well-known implementation method for
functional languages, with many variations: using so-called
super-combinators, etc. [see references in Lectures 5-6].

Now that we have an executable specification of the lambda
calculus as a rewrite theory, we can do the following in
the LAMBDA-2-CL system module below:

1.  Combine the lambda calculus and CL as subsorts
      of a supersort  MixedTerm containing both

2. Define the compilation of lambda expressions into CL
   by overloaded operators:

   op T : MixedTerm -> MixedTerm [frozen] .
   op T : Lambda{Vars} -> CL [frozen] .  

That is, T is defined on all MixedTerms and:
(i) is defined to be the identity on CL terms,
(ii) compiles any lambda term into a CL term.

Since the module LAMBDA-2-CL contains both calculi,
including their rules, but the sorts  CL and Lambda{Vars}
are disjoint [have no terms in common], we can execute
both lambda terms and CL terms without confusion; and we
can also: (i) compile a lambda term into a CL term, and
(ii) then execute the compiled CL term.

)



mod LAMBDA-2-CL is
  protecting LAMBDA-CALCULUS .
  sorts CL MixedTerm .
  subsorts CL Lambda{Vars} < MixedTerm .

*** this specifies CL within the LAMBDA-2-CL module
  
  ops S K I : -> CL [ctor] .
  op _ _ : CL CL -> CL [ctor gather (E e)] . ***
application

  vars E1 E2 E3 : CL .

  rl [K-red] : (K E1) E2 => E1 .
  rl [S-red] : ((S E1) E2) E3 => (E1 E3) (E2 E3) .
  rl [I-red] : I E1 => E1 .

*** end of CL specification

*** The MixedTerm sort builds "mixed terms" in the combined
*** language of the lambda-calculus and CL

  op \_._ : Vars MixedTerm -> MixedTerm [ctor] .
                            *** lambda abstraction
  op __ : MixedTerm MixedTerm -> MixedTerm
             [ctor gather (E e)] . *** application
  op fv : MixedTerm -> Set{Vars} [frozen] .

*** the fv function is extended to mixed terms

  vars X Y Z : Vars .  vars P Q : MixedTerm .

  eq fv(X) = X .
  eq fv(\ X . P) = fv(P) \ X .
  eq fv(P Q) = fv(P) + fv(Q) .
  eq fv(S) = mt .



  eq fv(K) = mt .
  eq fv(I) = mt .

*** Compiling lambda-expressions into CL

  op T : MixedTerm -> MixedTerm [frozen] .
  op T : Lambda{Vars} -> CL [frozen] .  *** translates
lambda to CL

  eq T(E1) = E1 .         *** T is the identity on CL
  eq T(P Q) = T(P) T(Q) .
  eq T(X) = X .
  ceq T(\ X . P) = K T(P)        if  X in fv(P) = false .
  eq T(\ X . X) = I .
  ceq T(\ X .(\ Y . P)) = T(\ X . T(\ Y . P))
                                 if  X in fv(P) = true .
  ceq T(\ X .(P Q)) = (S T(\ X . P)) T(\ X . Q)
                                 if  X in fv(P Q) = true .
endm

*** let us compile the composition operator:

red T(\ x{0} .(\ x{1} .(x{1} x{0}))) .

*** result CL: S (K (S I)) (S (K K) I)

rewrite (S (K (S I)) (S (K K) I) E1) E2 .

*** result CL: E2 E1

*** let us compile a \-term simulating K:

red T(\ x{0} .(\ x{1} . x{0})) .

*** result CL: S (K K) I



rewrite (S (K K) I E1) E2 .

*** result CL: E1

*** let us compile a \-term simulating S:

red T(\ x{0} .(\ x{1} . (\ x{2} . ((x{0} x{2}) (x{1}
x{2}))))) .

***(

result CL: S (S (K S) (S (K K) (S (K S)
(S (S (K S) (S (K K) I)) (K I)))))
(K (S (S (K S) (S (K K) I)) (K I)))

)

rewrite

((S (S (K S) (S (K K) (S (K S)
(S (S (K S) (S (K K) I)) (K I)))))
(K (S (S (K S) (S (K K) I)) (K I))) E1) E2) E3 .

*** result CL: E1 E3 (E2 E3)


